
**COMPUTER ORGANIZATION AND DESIGN**
The Hardware/Software Interface



Chapter 2

Instructions: Language of the Computer

(continued)

More Conditional Operations

- Set result to 1 if a condition is true
 - Otherwise, set to 0
- `slt rd, rs, rt`
 - if (`rs < rt`) `rd` = 1; else `rd` = 0;
- `slti rt, rs, constant`
 - if (`rs < constant`) `rt` = 1; else `rt` = 0;
- Use in combination with `beq`, `bne`
`slt $t0, $s1, $s2` # if (`$s1 < $s2`)
`bne $t0, $zero, L` # branch to L

Chapter 2 — Instructions: Language of the Computer — 2

Branch Instruction Design

- Why not `blt`, `bge`, etc?
- Hardware for `<`, `≥`, ... slower than `=`, `≠`
 - Combining with branch involves more work per instruction, requiring a slower clock
 - All instructions experience a penalty
- `beq` and `bne` are the common case (make the common case fast!)
- This is a good design compromise

Chapter 2 — Instructions: Language of the Computer — 3

Signed vs. Unsigned

- Signed comparison: `slt`, `slti`
- Unsigned comparison: `sltu`, `sltui`
- Example
 - `$s0 = 1111 1111 1111 1111 1111 1111 1111 1111`
 - `$s1 = 0000 0000 0000 0000 0000 0000 0000 0001`
 - `slt $t0, $s0, $s1 # signed`
 - $-1 < +1 \Rightarrow \$t0 = 1$
 - `sltu $t0, $s0, $s1 # unsigned`
 - $+4,294,967,295 > +1 \Rightarrow \$t0 = 0$

Chapter 2 — Instructions: Language of the Computer — 4

Calling a Procedure

§2.8 Supporting Procedures in Computer Hardware

- Steps required
 1. Place parameters in registers
 2. Transfer control to procedure
 3. Acquire additional storage for procedure
 4. Perform procedure operations
 5. Place result in register for caller
 6. Release additional storage for procedure
 7. Return to instruction directly after initial procedure call

Chapter 2 — Instructions: Language of the Computer — 5

Register Usage

- \$a0 - \$a3: arguments (reg's 4 - 7)
- \$v0, \$v1: result values (reg's 2 and 3)
- \$t0 - \$t9: temporaries
 - Can be overwritten by callee
- \$s0 - \$s7: saved
 - Must be saved/restored by callee
- \$gp: global pointer for static data (reg 28)
- \$sp: stack pointer (reg 29)
- \$fp: frame pointer (reg 30)
- \$ra: return address (reg 31)

Chapter 2 — Instructions: Language of the Computer — 6

PC: The Program Counter

- There is a special register called the *program counter* that holds the address of the current instruction
- Normally, this register is incremented by 4 each instruction
 - (remember that MIPS instructions are always 4 bytes each)
- When a branch happens - the **address** portion of the instruction is added to the PC register

Chapter 2 — Instructions: Language of the Computer — 7

The value of PC during instruction

- During the *execution* of an instruction, the processor always adds 4 to the PC register
- This happens *very early* in the instruction
- We should therefore assume that the PC always holds the address of the *next* instruction

Chapter 2 — Instructions: Language of the Computer — 8

Procedure Call Instructions

- Procedure call: jump and link
`jal ProcedureLabel`
 - Address of following instruction put in `$ra`
 - Jumps to target address
- Procedure return: jump register
`jr $ra`
 - Copies `$ra` to program counter
 - Can also be used for computed jumps
 - e.g., for case/switch statements

Chapter 2 — Instructions: Language of the Computer — 9

Leaf Procedure Example

- C code:

```
int leaf_example (int g, h, i, j)
{ int f;
  f = (g + h) - (i + j);
  return f;
}
```

 - Arguments `g, ..., j` in `$a0, ..., $a3`
 - `f` in `$s0` (hence, need to save `$s0` on stack)
 - Result in `$v0`
 - Note that a leaf here means that this procedure does not call any other procedures

Chapter 2 — Instructions: Language of the Computer — 10

Leaf Procedure Example

■ MIPS code:

leaf_example:		
addi	\$sp, \$sp, -4	Save \$s0 on stack
sw	\$s0, 0(\$sp)	
add	\$t0, \$a0, \$a1	Procedure body
add	\$t1, \$a2, \$a3	
sub	\$s0, \$t0, \$t1	
add	\$v0, \$s0, \$zero	Result
lw	\$s0, 0(\$sp)	Restore \$s0
addi	\$sp, \$sp, 4	
jr	\$ra	Return

Non-Leaf Procedures

- Procedures that call other procedures
- For nested call, caller needs to save on the stack:
 - Its return address
 - Any arguments and temporary registers needed after the call
- Restore from the stack after the call

Non-Leaf Procedure Example

- C code:

```
int fact (int n)
{
    if (n < 1) return 1;
    else return n * fact(n - 1);
}
```

 - Argument n in \$a0
 - Result in \$v0

Non-Leaf Procedure Example

- MIPS code:

fact:		
addi	\$sp, \$sp, -8	# adjust stack for 2 items
sw	\$ra, 4(\$sp)	# save return address
sw	\$a0, 0(\$sp)	# save argument
slti	\$t0, \$a0, 1	# test for n < 1
beq	\$t0, \$zero, L1	
addi	\$v0, \$zero, 1	# if so, result is 1
addi	\$sp, \$sp, 8	# pop 2 items from stack
jr	\$ra	# and return
L1:	addi \$a0, \$a0, -1	# else decrement n
	jal fact	# recursive call
lw	\$a0, 0(\$sp)	# restore original n
lw	\$ra, 4(\$sp)	# and return address
addi	\$sp, \$sp, 8	# pop 2 items from stack
mul	\$v0, \$a0, \$v0	# multiply to get result
jr	\$ra	# and return

Local Data on the Stack

The diagram illustrates the stack layout in three states: (a) initial state with `$fp` and `$sp` pointers, (b) state after saving registers and return address, and (c) state after allocating local data. The stack grows downwards from high to low addresses.

- Local data allocated by callee
 - e.g., C automatic variables
- Procedure frame (activation record)
 - Used by some compilers to manage stack storage

Chapter 2 — Instructions: Language of the Computer — 15

Memory Layout


The diagram shows the memory layout from high to low addresses: Stack, Dynamic data, Static data, Text, and Reserved. It also shows the addresses of `$gp`, `$pc`, and `$sp`.

- Text: program code
- Static data: global variables
 - e.g., static variables in C, constant arrays and strings
 - `$gp` initialized to address allowing \pm offsets into this segment
- Dynamic data: heap
 - E.g., malloc in C, new in Java
- Stack: automatic storage

Chapter 2 — Instructions: Language of the Computer — 16

Basic Block (BB)

- A *basic block* is a list of instructions to be executed in the given order
 - No embedded branches (except at end)
 - No branch targets (except at beginning)



- A compiler identifies basic blocks for optimization
- In loops, optimizations include loop-invariant code motion (LICM), strength reduction, etc.

Chapter 2 — Instructions: Language of the Computer — 17

Control Flow Graph (CFG)

- We build code by connecting the basic blocks together to form a directed graph called a *control flow graph* (CFG)
- All basic blocks have one entry, one exit
- Implies all instructions must execute in the given order and execute only once
- BB exit instructions are called *terminators*


Chapter 2 — Instructions: Language of the Computer — 18

Control Flow Graph (CFG)

- Graphical representation of program logic / flow
- Directed edges between basic blocks show *possible* code paths
- Any possible execution must be represented by a valid path
- In-edges from predecessors, out-edges to successors

Chapter 2 — Instructions: Language of the Computer — 19

CFG Example if / else




CFG for 'test_if' function

```

if ( <condition> )
{
    /* if True */
}
else
{
    /* if False */
}
/* merge here */
  
```

Chapter 2 — Instructions: Language of the Computer — 20

CFG Example for



CFG for loop example

 $j = 0$

```

while ( j < n )
{
    /* loop body */
    j++ /* latch */
}
  
```

```

/* loop exit */
  
```

Chapter 2 — Instructions: Language of the Computer — 21

Why Identify Basic Blocks?

- This is useful for structuring your assembly code
- It enables compilers to focus optimization efforts
 - 90% of runtime spent in 10% of code
 - Demonstrates the importance of loops

Chapter 2 — Instructions: Language of the Computer — 22

Character Data

§2.9 Communicating with People

- Byte-encoded character sets
 - ASCII: 128 characters
 - 95 graphic, 33 control
 - Latin-1: 256 characters
 - ASCII, +96 more graphic characters
- Unicode: 32-bit character set
 - Used in Java, C++ wide characters, ...
 - Most of the world's alphabets, plus symbols
 - UTF-8, UTF-16: variable-length encodings

Chapter 2 — Instructions: Language of the Computer — 23

Byte/Halfword Operations

- Could use bitwise operations
 - MIPS byte/halfword load/store
 - String processing is a common case
- `lb rt, offset(rs)` `lh rt, offset(rs)`
 ■ Sign extend to 32 bits in `rt`
`lbu rt, offset(rs)` `lhu rt, offset(rs)`
 ■ Zero extend to 32 bits in `rt`
`sb rt, offset(rs)` `sh rt, offset(rs)`
 ■ Store just rightmost byte/halfword

Chapter 2 — Instructions: Language of the Computer — 24

String Copy Example

- C code:
 - Relies on null-terminated string
- ```
void strcpy (char * x, char * y)
{ int i;
 i = 0;
 while ((x[i]=y[i])!='\0')
 i += 1;
}
```
- Addresses of x, y in \$a0, \$a1
  - i in \$s0

# String Copy Example

- MIPS code:

|         |                      |                           |
|---------|----------------------|---------------------------|
| strcpy: |                      |                           |
| addi    | \$sp, \$sp, -4       | # adjust stack for 1 item |
| sw      | \$s0, 0(\$sp)        | # save \$s0               |
| add     | \$s0, \$zero, \$zero | # i = 0                   |
| L1:     | add \$t1, \$s0, \$a1 | # addr of y[i] in \$t1    |
| lbu     | \$t2, 0(\$t1)        | # \$t2 = y[i]             |
| add     | \$t3, \$s0, \$a0     | # addr of x[i] in \$t3    |
| sb      | \$t2, 0(\$t3)        | # x[i] = y[i]             |
| beq     | \$t2, \$zero, L2     | # exit loop if y[i] == 0  |
| addi    | \$s0, \$s0, 1        | # i = i + 1               |
| j       | L1                   | # next iteration of loop  |
| L2:     | lw \$s0, 0(\$sp)     | # restore saved \$s0      |
| addi    | \$sp, \$sp, 4        | # pop 1 item from stack   |
| jr      | \$ra                 | # and return              |

§2.10 MIPS Addressing for 32-Bit immediates and Addresses

## 32-bit Constants

- Most constants are small
  - 16-bit immediate is sufficient
- For the occasional 32-bit constant
  - `lui rt, constant`
    - Copies 16-bit constant to left 16 bits of `rt`
    - Clears right 16 bits of `rt` to 0

`lui $s0, 61`

0000 0000 0111 11010000 0000 0000 0000

`ori $s0, $s0, 2304`

0000 0000 0111 11010000 1001 0000 0000

Chapter 2 — Instructions: Language of the Computer — 27

## Branch Addressing

- Branch instructions specify
  - Opcode, two registers, target address
- Most branch targets are near the branch
  - Make the common case fast!


|        |        |        |                     |
|--------|--------|--------|---------------------|
| op     | rs     | rt     | constant or address |
| 6 bits | 5 bits | 5 bits | 16 bits             |

- PC-relative addressing
  - Target address =  $PC + \text{offset} \times 4$
  - PC already incremented by 4 by this time

Chapter 2 — Instructions: Language of the Computer — 28

# Jump Addressing

- Jump (j and jal) targets could be anywhere within the text segment
  - Encode full address in instruction



- (Pseudo)Direct jump addressing
  - Target address =  $PC_{31...28} : (address \times 4)$

# Target Addressing Example

- Loop code from earlier example
  - Assume Loop at location 80000

|           |                  |       |    |                         |    |                 |   |    |
|-----------|------------------|-------|----|-------------------------|----|-----------------|---|----|
| Loop: sll | \$t1, \$s3, 2    | 80000 | 0  | 0                       | 19 | 9               | 4 | 0  |
| add       | \$t1, \$t1, \$s6 | 80004 | 0  | 9                       | 22 | 9               | 0 | 32 |
| lw        | \$t0, 0(\$t1)    | 80008 | 35 | 9                       | 8  | 0               |   |    |
| bne       | \$t0, \$s5, Exit | 80012 | 5  | 8                       | 21 | 2 (i.e., 4 * 2) |   |    |
| addi      | \$s3, \$s3, 1    | 80016 | 8  | 19                      | 19 | 1               |   |    |
| j         | Loop             | 80020 | 2  | 20000 (i.e., 4 * 20000) |    |                 |   |    |
| Exit: ... |                  | 80024 |    |                         |    |                 |   |    |


# Branching Far Away

- If branch target is too far to encode with 16-bit offset, assembler rewrites the code


- Example

```
beq $s0,$s1, L1
 ↓
bne $s0,$s1, L2
j L1
L2: ...
```

# Addressing Mode Summary







## Assembler Pseudoinstructions

- Most assembler instructions represent machine instructions one-to-one
- Pseudoinstructions: figments of the assembler's imagination
 

`move $t0, $t1`     $\rightarrow$  `add $t0, $zero, $t1`  
`blt $t0, $t1, L`    $\rightarrow$  `slt $at, $t0, $t1`  
                               `bne $at, $zero, L`
- `$at` (register 1): assembler temporary

Chapter 2 — Instructions: Language of the Computer — 34

## Producing an Object Module

- Assembler (or compiler) translates program into machine instructions
- Provides information for building a complete program from the pieces
  - Header: described contents of object module
  - Text segment: translated instructions
  - Static data segment: data allocated for the life of the program
  - Relocation info: for contents that depend on absolute location of loaded program
  - Symbol table: global definitions and external refs
  - Debug info: for associating with source code

Chapter 2 — Instructions: Language of the Computer — 35

## Linking Object Modules

- Produces an executable image
  1. Merges segments
  2. Resolve labels (determine their addresses)
  3. Patch location-dependent and external refs
- Could leave location dependencies for fixing by a relocating loader
  - But with virtual memory, no need to do this
  - Program can be loaded into absolute location in virtual memory space

Chapter 2 — Instructions: Language of the Computer — 36

## Loading a Program

- Load from image file on disk into memory
  1. Read header to determine segment sizes
  2. Create virtual address space
  3. Copy text and initialized data into memory
    - Or set page table entries so they can be faulted in
  4. Set up arguments on stack
  5. Initialize registers (including \$sp, \$fp, \$gp)
  6. Jump to startup routine
    - Copies arguments to \$a0, ... and calls main
    - When main returns, do exit syscall

Chapter 2 — Instructions: Language of the Computer — 37

## Dynamic Linking

- Only link/load library procedure when it is called
  - Requires procedure code to be relocatable
  - Avoids image bloat caused by static linking of all (transitively) referenced libraries
  - Automatically picks up new library versions

Chapter 2 — Instructions: Language of the Computer — 38

# Lazy Linkage

Indirection table

Stub: Loads routine ID, Jump to linker/loader

Linker/loader code

Dynamically mapped code

Text

jal

lw

jr

Data

Text

ld

Text

Dynamic linker/loader

Remap DLL routine

Data/Text

DLL routine

Text

jal

lw

jr

Data

Text

DLL routine

a. First call to DLL routine

b. Subsequent calls to DLL routine

Chapter 2 — Instructions: Language of the Computer — 39

# Starting Java Applications

Java program

Compiler

Class files (Java bytecodes)

Just In Time compiler

Compiled Java methods (machine language)

Simple portable instruction set for the JVM

Java library routines (machine language)

Java Virtual Machine

Interprets bytecodes

Compiles bytecodes of "hot" methods into native code for host machine

Chapter 2 — Instructions: Language of the Computer — 40

§2.13 A C Sort Example to Put It All Together

## C Sort Example

- Illustrates use of assembly instructions for a C bubble sort function
- Swap procedure (leaf)

```
void swap(int v[], int k)
{
 int temp;
 temp = v[k];
 v[k] = v[k+1];
 v[k+1] = temp;
}
```

  - v in \$a0, k in \$a1, temp in \$t0

Chapter 2 — Instructions: Language of the Computer — 41

## The Procedure Swap

|                         |                             |
|-------------------------|-----------------------------|
| swap: sll \$t1, \$a1, 2 | # \$t1 = k * 4              |
| add \$t1, \$a0, \$t1    | # \$t1 = v+(k*4)            |
|                         | # (address of v[k])         |
| lw \$t0, 0(\$t1)        | # \$t0 (temp) = v[k]        |
| lw \$t2, 4(\$t1)        | # \$t2 = v[k+1]             |
| sw \$t2, 0(\$t1)        | # v[k] = \$t2 (v[k+1])      |
| sw \$t0, 4(\$t1)        | # v[k+1] = \$t0 (temp)      |
| jr \$ra                 | # return to calling routine |

.align 2      # word alignment, 2^2

Chapter 2 — Instructions: Language of the Computer — 42

# The Sort Procedure in C

- Non-leaf (calls swap)

```
void sort (int v[], int n)
{
 int i, j;
 for (i = 0; i < n; i += 1) {
 for (j = i - 1;
 j >= 0 && v[j] > v[j + 1];
 j -= 1) {
 swap(v,j);
 }
 }
}
```
- v in \$a0, k in \$a1, i in \$s0, j in \$s1

# The Procedure Body

|                               |                                      |                          |
|-------------------------------|--------------------------------------|--------------------------|
| move \$s2, \$a0               | # save \$a0 into \$s2                | Move<br>params           |
| move \$s3, \$a1               | # save \$a1 into \$s3                |                          |
| move \$s0, \$zero             | # i = 0                              | Outer loop               |
| for1tst: slt \$t0, \$s0, \$s3 | # \$t0 = 0 if \$s0 ≥ \$s3 (i ≥ n)    |                          |
| beq \$t0, \$zero, exit1       | # go to exit1 if \$s0 ≥ \$s3 (i ≥ n) |                          |
| addi \$s1, \$s0, -1           | # j = i - 1                          |                          |
| for2tst: slti \$t0, \$s1, 0   | # \$t0 = 1 if \$s1 < 0 (j < 0)       |                          |
| bne \$t0, \$zero, exit2       | # go to exit2 if \$s1 < 0 (j < 0)    |                          |
| sll \$t1, \$s1, 2             | # \$t1 = j * 4                       | Inner loop               |
| add \$t2, \$s2, \$t1          | # \$t2 = v + (j * 4)                 |                          |
| lw \$t3, 0(\$t2)              | # \$t3 = v[j]                        |                          |
| lw \$t4, 4(\$t2)              | # \$t4 = v[j + 1]                    |                          |
| slt \$t0, \$t4, \$t3          | # \$t0 = 0 if \$t4 ≥ \$t3            |                          |
| beq \$t0, \$zero, exit2       | # go to exit2 if \$t4 ≥ \$t3         |                          |
| move \$a0, \$s2               | # 1st param of swap is v (old \$a0)  | Pass<br>params<br>& call |
| move \$a1, \$s1               | # 2nd param of swap is j             |                          |
| jal swap                      | # call swap procedure                |                          |
| addi \$s1, \$s1, -1           | # j -= 1                             | Inner loop               |
| j for2tst                     | # jump to test of inner loop         |                          |
| exit2: addi \$s0, \$s0, 1     | # i += 1                             | Outer loop               |
| j for1tst                     | # jump to test of outer loop         |                          |

# The Full Procedure

|       |                     |                                      |
|-------|---------------------|--------------------------------------|
| sort: | addi \$sp,\$sp, -20 | # make room on stack for 5 registers |
|       | sw \$ra, 16(\$sp)   | # save \$ra on stack                 |
|       | sw \$s3,12(\$sp)    | # save \$s3 on stack                 |
|       | sw \$s2, 8(\$sp)    | # save \$s2 on stack                 |
|       | sw \$s1, 4(\$sp)    | # save \$s1 on stack                 |
|       | sw \$s0, 0(\$sp)    | # save \$s0 on stack                 |
|       | ...                 | # procedure body                     |
|       | ...                 |                                      |
| exit: | lw \$s0, 0(\$sp)    | # restore \$s0 from stack            |
|       | lw \$s1, 4(\$sp)    | # restore \$s1 from stack            |
|       | lw \$s2, 8(\$sp)    | # restore \$s2 from stack            |
|       | lw \$s3,12(\$sp)    | # restore \$s3 from stack            |
|       | lw \$ra,16(\$sp)    | # restore \$ra from stack            |
|       | addi \$sp,\$sp, 20  | # restore stack pointer              |
|       | jr \$ra             | # return to calling routine          |